

Fichier tableur "sonde_phoenix"

Formules des cellules

(Formules pouvant être utilisées par copié-collé)

I - Feuilles *Ephem Terre* et *Ehem Mars*

Extraction des données

date	=DATE(STXT(A16;9;4);STXT(A16;5;2);STXT(A16;1;3))
X planète	=CNUM(STXT(A16;27;17))
Y planète	=CNUM(STXT(A16;45;17))
Z planète	=CNUM(STXT(A16;63;17))
dist. au Soleil	=CNUM(STXT(A16;82;17))

Longitudes et latitudes

long. écliptique	=MOD(DEGRES(ATAN2(C16;D16));360)
lat. écliptique	=DEGRES(ASIN(E16/F16))

II - Feuille *Orbites*

Test pour trajectoire de la Terre =SI(E17<E16;"-";"+")

Test pour trajectoire de Mars =SI(J17<J16;"-";"+")

Tableau I - Périhélies et aphélies de la Terre et Mars

Longitude du périhélie	=MOD(DEGRES(ATAN2(V60;W60));360)
Demi grand axe	=(P60+U60)/2
excentricité	=(Y60-U60)/Y60

Tableau II - Positions des planètes au départ et arrivée

Terre

demi grand axe	=(E50+E51)/2
longitude du périhélie	=MOD(DEGRES(ATAN2((B50+B51)/2;(C50+C51)/2))+360;360)

Mars

demi grand axe	=(J50+J51)/2
longitude du périhélie	=MOD(DEGRES(ATAN2((G50+G51)/2;(H50+H51)/2))+360;360)
longitude à l'arrivée de la sonde	=MOD(O68+180;360)
distance au Soleil à l'arrivée	='[sonde_phoenix_travail]Ephem Mars'!F277

Tableau III - Orbite de la sonde

Demi grand axe a	=(N68+Q69)/2
Période P	=RACINE(O75*O75*O75)*an "an" nom de la cellule H5
excentricité	=(O75-N68)/O75
Durée du voyage	=O76/2
Date d'arrivée	=A50+O78
Long. du périhélie	=O68
Long. de l'aphélie	=P69
Angle b0	=O69-O68+360

Orbite de la sonde

angle θ	=AB16*180/np "np" cel. AF12 nbre de points du tracé
r sonde	=aa*(1-e*e)/(1+e*COS(RADIANS(AC16))) "aa" cel. AF9
X sonde	=AD16*COS(RADIANS(MOD(AC16+b0;360))) "b0" cel. AF11
Ysonde	=AD16*SIN(RADIANS(MOD(AC16+b0;360)))

Tableau IV - Vitesses moyennes des planètes

Vitesse moyenne	=2*PI()*(O88/(Q88*24*3600))*(1-P88^2)^(-0.5)
Vitesse périhélie	=R88*(1+P88)
Vitesse aphélie	=R88*(1-P88) Mars indice ligne 89, Sonde indice ligne 90